

JMOL メモ

Jmol <http://jmol.sourceforge.net/>

高機能な分子表示 (エディットも可) プログラムです。オープンソース・フリーソフトウェアでありライセンスは LGPL。Win,Mac,Linux 版があります。Java で作成されているので Java の実行環境が必要です。JavaScript 版もあり Web 上での操作もできます。蛋白などの大きな分子の表示機能も豊富で、多くの MO 計算プログラムの計算結果を可視化できます。継続的に開発が続けられています。



An OpenScience project
©1999-2015 by the Jmol Team
Jmol is free, opensource software

Jmol is an Open Source molecular viewer licensed under the GNU Lesser General Public License.

画面上方のメニューと画面上の右クリックから機能呼び出しをすることができますが、コンソールにコマンドを打ち込んでの操作や、スクリプトエディタ上で複数のコマンドを一度に実行することにより、迅速で効率的な操作を行うことができます。これらを組み合わせて使うこともできます。コマンドの一部がメニューから実行できるようになっています。

- 操作法 1 : マウス操作、ボタンやメニュー (使用できる機能は少ない)、右クリックメニューによる操作 (機能は中程度)
- 操作法 2 : コンソール上にキーボードでコマンドを打ち込むことによる操作 [ファイル]→[コンソール] (全機能が使用可)
- 操作法 3 : スクリプトエディタでコマンドを一度に実行 (プログラミングによるループ処理などで操作の自動化も可能) [ファイル]→[スクリプトエディタ]

起動 (PC,Mac では jmol.bat, jmol.jar 等を実行) (WS 教室では、ターミナルから **jmol&**と入力)


ファイルの読み込み 開いたウィンドウに、分子の座標ファイル、Gaussian 等のログファイルをドロップすることで読み込むことができます。メニューの [ファイル] → [開く]、コンソールから load "ファイル名" でも同様。非常に多くのファイル形式に対応しています。

PDB(Protein Data Bank)の分子 ID がわかる場合には、[ファイル] → [PDB を取得] によりネット経由で取得可。


(注 : Gaussian 等のログファイルを読み込んだ場合、このとき表示された構造は、収束構造ではなく、初期構造等の log ファイル中にある最初の座標で、必要に応じて使用する構造を選択する必要があります。)

The screenshot shows the Jmol software interface with a menu bar (File, Edit, View, Tools, Macro, Help) and a toolbar. Numerous annotations in Japanese provide detailed explanations of the software's features:

- Toolbar:**
 - Image as save (imageとして保存)
 - Save current as script (現状をスクリプトとして保存)
 - Shift-drag to select and measure distances (シフトドラッグで囲んで選択 距離等を測定)
 - Center (中心に)
 - Home position (ホームポジションに)
 - Model selection (model の選択)
- File Menu:**
 - File creation mode (分子の作成モードであることを表す)
 - Save as image (画像をコピー)
 - Environment settings (環境設定) and atom sizes (原子サイズ等)
 - Script editor (スクリプトエディタ) for command input (コマンドの入力)
 - Batch execution of commands (コマンドの一括実行) using variables, loops, and if-else statements (変数や for ループ、if 分岐を使ったプログラムとしての実行)
- View Menu:**
 - Van der Waals radii (van der Waals半径) settings: 15%, 20%, 25%, 100%
 - Wireframe (ワイヤフレーム) options
 - Scale (スケール) settings: 0.2, 0.5, 1, 2, 5
- Tools Menu:**
 - Measurement tools (計測)
 - Animation (アニメーション)
 - Selection tools (選択ツール)
 - Gaussian input file creation (Gaussianの入力ファイルの作成)
 - Frame management (フレーム管理)
- Right-click context menu:**
 - MMFF94, UFF force field calculations (MMFF94, UFF 力場による分子力学計算)
 - File output in MOL or XYZ format (座標ファイルへの出力は MOL 形式 XYZ 形式)
 - Return to original view (右クリックの表示を元に戻す [作成モードの終了])
- Other annotations:**
 - Reverse secondary elements (四級炭素等をクリックして反転)
 - Other elements are entered manually (他の元素は元素記号を手入力する)
 - Jmol has a one-time molecule model creation function, but Avogadro is recommended (Jmol には一通りの分子モデル作成機能がありますが、Avogadro の方が使いやすくお勧めです。)



分子構造の読み込み方法

画面上に座標ファイルをドロップする。または、ファイル→開く、または、 から。

ファイル→URL を指定して読み込む。

ファイル→Protein DataBank から PDB ID(数字と3 英文字)や CIF ファイル名等を与えて読み込む。コマンドの場合は、=5GIT 等
ファイル→Mol を取得から読み込む。

その他コマンドで3文字の化合物名で load=hem 名称は、<http://ligand-expo.rcsb.org/> で検索
PubChem のデータベースから load: 例 load:TMU (TetraMethylUrea)
load "ファイル名"

コマンドでファイルを開くには、

パスの指定がないと Jmol.jar の存在するフォルダから読み込まれるので、パスの変更を行っておくと特定のフォルダから読み込める。
読み込むファイルのあるディレクトリを指定 set defaultdirectory "パス"
書き出すディレクトリを指定 cd "パス"

Jmol の詳細なコマンド一覧 <https://chemapps.stolaf.edu/jmol/docs/>

コマンドをキーボードから入力するコンソールの開き方

メニューの [ファイル] → [コンソール] または、画面上で 右クリック → [コンソール]


過去の入力は、上矢印キー ↑ で呼び出せる。

コマンドをまとめてコンソールにペーストしてまとめて入力することも可

コマンドは、Undo/Redo ができる undo / redo

コマンドの履歴を表示 show history

[ファイル] → [スクリプトエディタ] にコマンドを入力して [実行] によりプログラムとしても実行できる。

現在の表示状態を再現するスクリプトを得るには、左から5番目のボタン  [現在の状況を書き出し]

または、メニューの File→Export→Write State... からスクリプトとして保存することができる。(かなりの行数)

この中身をスクリプトエディタで実行すると再現できる。

コマンドは ; で1行に続けることができる。 コメント行は、#

全ての座標や変数の値を初期化する reset (zoom 100; center; translate x 0; translate 0)

画面右下の "Jmol" ログを隠す/表示する

コンソールから set frank off/on



言語インターフェースの変更

右クリック [言語] → [各言語]

コンソールから set language en / jp

英語 / 日本語 モードに

画面のサイズの調整

画面下中央に縦横のサイズが表示されています。これが書き出される際の画像ファイルのサイズとなる。

ウィンドウの右下隅をドラッグして調整

メニューの [表示] → [サイズ変更] から数値で与える。

640 x 480

set window 640 400

画面の画像を書き出す (スクリーンショットの方が容易です)

メニューから [ファイル] → [エクスポート] → [画像として書き出し] (JPEG 形式)

右クリック [ファイル] → [Export] (JPEG, GIF, PNG 形式他)

write image "ファイル名"


JPEG 形式で出力

write image png "ファイル名"

PNG 形式で出力

write image pngT "ファイル名"

背景を透明とした PNG 形式で出力



分子表示の奥行きのある/なし

メニューから、[表示] → [透視深度]

右クリックから [スタイル] → [視点の奥行き]

コンソールから、

set perspectiveDepth on / off

奥行き感のある/なし

set cameraDepth 0.1

深度をかなり大きく (右図). (3.0 がデフォルト値)

Z 方向の奥行をぼかす/消す

set zShade on/off

奥を見えにくくする

set zShadePower 1~10

奥を隠す (見えなくする). 1 は隠さない


グラフィックスを (中間色を使って) 滑らかに表示 (アンチエイリアス表示) する。

set antialiasDisplay ON/OFF

画面表示を滑らかにする

set antialiasImage True/False

出力されるファイルの画像を滑らかにする



分子の移動

control + 右クリックでマウスを移動


コンソールから

translate x 10

X 軸方向に 10 移動. Y, Z も可

translateSelected y 100

選択されている分子のみの移動 translate Selected とスペースを入れても良い



分子の回転

Z 軸まわりの回転

shift + 右クリックでマウスを移動

shift + 左クリックで左右にマウスを移動


コンソールから

rotate z 90

Z 軸方向に 90° 回転. Y 軸, Z 軸も可

rotateSelected y 20

選択されている分子のみの回転 rotate Selected も可



rotate selected best
show orientation

選択した分子を最適な向きにする。 rotate best も可
現在方向まで動かす (拡大等) 命令が表示される。(これを使うと同じ位置を再現できる)
出力の前半には、アニメーションで移動の場合があり、後半は移動のみを行う命令がある。
現在位置から初期位置まで動かす命令が表示される。


show moveto

分子の拡大/縮小

マウスのホイール または shift + 左クリックで上下にマウスを移動


コンソールから

zoom 200 2倍のサイズに拡大
zoom +100 同上
zoom *2 同上
show zoom 現在の拡大縮小度合いを表示



指定した原子を画面の中心とする

center _Cu

Cu 原子を画面の中心に移動する。  ボタンでも可
クリックした原子が画面の中心に移動する。

center selected

選択された原子を中心にする。複数原子を指定した場合には、重心が中心となる。

特定の原子、残基等を選択する (初期状態では全原子が選択されている) (選択されている原子に対して表示方法の変更等ができる)

選択された原子を丸で囲んで表示する

右クリック → [選択対象を暈(うん)で囲む]

selectionHalos ON/OFF




右クリックから [選択] → [すべて/なし, 反転, 元素, タンパク質, 核酸, ヘテロ 等]



を使う (シフトキーを押しながら矩形選択) select none の後に使うのが適当。

コンソールから

select all / none 全選択/非選択 右クリック→選択→すべて/なし
select on / off 選択の一時的 ON/OFF
select not selected 選択領域の反転 右クリック→選択→反転
原子の指定は、元素名、元素記号、番号の範囲等で可。
select _O 酸素原子全てを選択
select oxygen 同上
select _O , _H 酸素と水素を選択。コンマで区切る
select @1 番号1の原子を選択する。
select atomNo=1 同上
select @1,@2 番号1と4の原子を選択する。カンマで並べる
select atomNo=1, atomNo=4 同上
select atomNo>=2 AND atomNo<=4 番号2,3,4の原子を選択する。数値の範囲で指定する。ORも可
select (C1,C2) 番号1,2の炭素を選択する。
select add _H 追加選択
select remove _H 選択領域から引く
select within(3.0, _Cu) Cuから距離3.0Å以内の原子を選択する
select water 水分子を選択する。(PDBの場合はHOHの場合もあり)
select G / C / A / T / U 核酸を選択 (PDB等のファイル内で定義されている場合)
select selected & _AI 選択ツール  で範囲を選択してから特定の原子を選択する

選択原子をわかりやすく表示する。

star on 選択原子に+印を出す 他方法として cpk 100% 等で大きく表示してもよい
color star red 赤くする
set starWidth 0.2 太くする
star off

display _C

炭素原子のみを表示する。 undo で戻せます。全て表示は display all

hide _C

炭素原子を隠す

select hidden, select not hidden, hide hidden, hide not hidden などが可

InvertedSelected

選択部分を幾何的中心で反転させる。中心を与えたり、面を与えることもできる

分子の表示方式を変更するには、

メニューからは、[表示] → [原子] で半径のみの変更が可能。

右クリックでは [スタイル] → [原子] 同 → [結合] 同 → [ラベル]

[スタイル] → [モデルの表示方法] → [CPK 空間充填] [棒球] [棒] [ワイヤーフレーム] [カートゥーン] [トレース]

コンソールから原子サイズを変更するには、

cpk on/off cpk 表示の on/off (cpk は、spacefill でもよい)
cpk 100% cpk 表示の比率を変更する (初期サイズに比例して変更できる) 初期値 20%
cpk -200~749 cpk 表示を数値で指定 (全ての原子が同じサイズとなる)
cpk ionic イオン半径表示にする
color atoms TRANSLUCENT 0.5 半透明にする 0 が不透明、0.99 がほとんど透明

結合の太さを変更する



wireframe on/off
 wireframe 0~750 の数値
 set bonds on/off

比率の % を付けての設定も可
 二重結合等を表示するか否か

色の指定 (選択された原子に対して) Jmol の色指定の一覧 <http://jmol.sourceforge.net/jscolors/>

右クリックから [色] → [原子、結合、構造 等]
 蛋白の場合 [色] → [構造] → [カートゥーン] → [スキームで] → [基]


コンソールから

color red 選択部分を赤く。色名は上記の URL に多数あり
 RGB の 10 進数で指定する場合 [255,0,0]、16 進数で指定する場合[xFF0000]または"#FF0000"
 Jmol のデフォルト色とする。color cpk も同様

color jmol 蛋白等の場合 color cpk / amino / shapely / structure / group / chain 等が可
 原子と結合を異なる色にしたい場合は、
 color atom 色
 color bond 色
 color bonds none 結合の色をデフォルト設定 (原子の色) にする
 select _C; set bondmode OR; color bonds blue 炭素原子からの結合の色を全て青に
 select _C; set bondmode AND; color bonds blue C-C 結合のみの色を青に

半透明色にしたい場合

color 文の後ろに translucent 0.5 等を指定する。0 が不透明、1.0 が透明。原子、結合等に対して指定可能。
 影がなくなるので立体感がなくなる color translucent 0.5 または、color red translucent 0.5 等



背景色の指定

右クリックから [色] → [背景] → [白] 等

コンソールから

background white 背景色を白に。色の指定は上記と同様に色名や RGB の数値も可


各原子に元素名、番号などを表示するには、

メニューから、[表示] → [ラベル]

右クリックから、[スタイル] → [ラベル] → [元素記号、原子名、原子番号、表示位置]

コンソールから、

label on/off 元素記号を表示
 label %e 原子番号を表示
 label %i 原子記号と番号を表示
 label %a 原子記号と番号を表示
 set fontSize 24 フォントサイズを 24 ポイントに
 set labelOffset 0 3 ラベルの表示位置を真上に(dx=0, dy=3)に
 set labelFront on/off 他の原子の下側になったときに隠すか
 color label blue ラベルの色を青に



図の中にタイトル等の文字を表示する

画面の下中央に、32point のゴシック体で、青色で "図 1" と表示する場合の例。


set echo bottom center 上下の指定は top / middle / bottom, 左右は left / center / right
 font echo 32 sanszSerif ゴシックの 32 ポイントを設定 (明朝系フォントは serif)
 color echo blue
 echo 図 1



水素結合の算出と表示

右クリックから [スタイル] → [水素結合] → [計算する]、オン/オフ 等

hbonds calculate (距離と角度のみから)水素結合を算出する。
 hbonds 50 ボンドの表示太さの変更
 hbonds off / on / delete 表示のオフ/オン/削除
 set hbond solid 点線でなく実線で描く



距離・角度・二面角の測定と、測定結果の分子上への表示

- ボタンから
- メニューの [ツール] → [計測] から
- 右クリックで [計測値]から
- コンソールから

measure 1 2 3 4 番号 1, 2, 3, 4 原子の二面角を表示する。
 measure delete 消す
 measure on/off 表示する/しない
 font measurements 24 フォントサイズを 24 ポイントにする



font measurements 24

長さの単位の変更


メニューの [ツール] → [長さの単位] から

右クリックで [計測]→[長さの単位ナノメートル/オングストローム] 等

コンソールから

set measurementUnits angstroms / nm / pm

線を太くする (右図のように)



→ `set measurements 0.03 #単位Å` または `set measurements 3 #単位ピクセル`
 測定値の文字と線の色を赤に変更する `color measures red`

さらに他の色に変更するには、`color measures none` で元に戻してから変更する。



測定値のリストを表示する `measure list`

特定の原子@1 の表示を大きくして、@1 から 3.0Å 以内の全ての 0 原子までの結合距離を表示する

```
select none
select @1
cpk 20%
select none
select within(3.0, @1)
hide not selected
measure @1 {_O} all "%0.3VALUE"
```

二面角の表示例

マウスで、結合角や二面角を直接測定する場合の操作上の注意点 (二面角を例に)



ウィンドウ上中央の測定用ボタンを”押さない”状態で、(測定ボタンなどが押された状態で)

- 1つ目の原子を番号がポップアップしてからダブルクリック、
- 2つ目の原子を 同上 クリック、 (このとき、赤色で結合距離が表示されている)
- 3つ目の原子を 同上 クリック、 (このとき、赤色で結合角が表示されている)
- 4つ目の原子を 同上 ダブルクリック (赤色の二面角が、黒色となって確定する)

測定後は、余白をクリックで解除して (解除できていないときは右クリック等) 再測定に進む。

2つ目や3つ目の原子が正確にクリックされていないと二面角が表示されず、角度が表示される等となる場合がある。

3つ目の原子をダブルクリックすると角度の測定となる。

ウィンドウ上中央の測定用ボタンを使った場合には、これが選択された状態では角度や二面角を測定できないので、[分子を回転させる]  ボタンを選んでから再測定する。

原子間に線を引く 番号1と2の原子間に line1 という名前を付けて赤色で線を引く

```
draw line1 @1 @2 color red
```

分子力学法による構造最適化、チャージを与える、省略されている水素を付加する等

右クリックから [計算] → [構造最適化]


コンソールから

<code>minimize</code>	構造最適化を開始
<code>minimize energy</code>	最適化は行わず、現在の構造のエネルギーのみを算出
<code>set forceField "MMFF94"</code>	MMFF94 力場を使う。パラメータのない原子があった場合には全体に UFF が適用される。
<code>set forceField "UFF"</code>	UFF 力場を使う。(全原子のパラメータあり。精度低め)
その他、一部分の構造固定による最適化なども可	
<code>calculate FormalCharge</code>	形式電荷を与える
<code>color FormalCharge</code>	原子の色を形式電荷とする。
<code>calculate PartialCharge</code>	MMFF94 によるチャージを与える
<code>calculate hydrogens</code>	水素を付加する。水素が省略されたカルボン酸の O-C=O は、HO-C-OH となることがあるので注意。 COO ⁻ の場合も同様。
<code>set pdbAddHydrogens True</code>	PDB ファイルの読み込み時に自動的に水素を付加する。

分子表面の表示

コンソールから

前述 `cpk on / off` , `cpk 100%` , `spacefill on / off` , 球面で表示
`dots on / off` ドット表示の表面
`color dots red` ドットの色を赤に
`dots 1.0 / 0.2 / 2.0` 等 整数 2 は不可 ドット表示面の球のサイズを指定 (1.0 Å)
`dots VANDERWAALS` vdW 半径でドットを表示
`set dotDensity 3` ドットの密度。3 が最大




溶媒接触表面等の描画

右クリック→面

コンソールから

半径 1.0 Å の球が接触する分子の表面自体を描く `isoSurface solvent 1.0` (左図)
 接触する半径 1.0 Å の球の中心が描く表面 `isoSurface saSurface 1.0` (右図)
 表面のオン/オフ `isoSurface on / off` 左右図は同一スケール
 半透明の表面とする `isosurface saSurface 1.0 translucent 0.5` (同 左図) `color isosurface yellow`
 (上記で内部が表示されないとき `wireframe 0.2` 等で結合を太くして見えるようにする等)




分子の断面を表示：コンソールから、`slab on/off`

`Control+Shift` 上下にドラッグすると断面が現れる `slab 50` でも可(0~100)


分子モデル、軌道のいずれの断面も表示できる。 右図：C60 の断面

立体視 右クリック→[スタイル]→[立体視の方式]→[立体視眼鏡、交差法、平行法] 等

コンソールから `stereo` , `stereo 5` (デフォルト値)



アニメーション表示と GIF アニメファイル化 IR 振動のアニメーションは別記

右クリック→ [回転]
 右クリック→ [アニメーション]
 spin z 180 Z 軸方向に 180° /sec で回転のアニメーションを行う。(中心の指定は center  で)
 spin [_Cu] y 180 Y 軸方向に Cu 原子を中心として回転のアニメーションを行う
 spin (atomNo=1) (atomNo=2) 180 番号 1 と 2 を軸として回転
 spin on / off 回転を開始/停止
 reset spin 回転開始以前の位置に戻す
 zoomTo 2 200 2 秒で 2 倍のサイズに拡大するアニメーション表示
 zoomTo 2 [_Cu] 200 Cu 原子を中心として.. 同上 (画面の中心にない場合は移動して)

移動回転拡大を同時に行うアニメーション

move 360 0 0 0 0 0 3 3 秒かけて X 軸を 360 度回転
 move -990 0 0 150 30 -20 0 0 3 reset spin で元の位置に戻ります, undo でも可
 回転角度 x y z ズーム 移動 x y z 秒数 途中の 0 は slab の意味で現在動作せず無関係.

GIF(PNG)アニメファイルの作成 (最終ページに作成例)


開始/終了を指定してその間の表示コマからアニメーションを作成
 capture "test.gif" 以降の動作をアニメーション GIF ファイルに出力する。(マウスで分子を動かしても良い)
 capture end 右クリック: ファイル→Capture→Start capturing
 capture off/on ファイル出力を終了し保存する。 右クリック: ファイル→Capture→End capturing
 animation FPS 10 ファイル出力を一時的に停止/再開する 右クリックでも可
 animation Mode Loop フレーム速度の指定 10 フレーム/秒とする 右クリックでも可
 時間を指定してアニメーションを作成 (記録時間中にマウスで分子を動かしても良い) ループ再生の GIF アニメとする. 1 回で停止する場合は, Once 右クリックでも可
 capture "test.gif" 秒数 動作を指定した秒数だけアニメーション GIF ファイルに出力する。
 capture Transparent "test.gif" 秒数 同、背景色を透明として出力する。
 capture "ファイル名.png" 秒数 動作をコマ毎に別々の png ファイルに書き出す,
 capture on / off 時間を与えるのではなく静止画のコマから作成するには, on 後にコマを表示して off する。
 回転と同時にファイル作成させる X 軸方向に回転させながら GIF ファイルを作成
 capture "test.gif" spin Y

2つの pdb ファイルの線形補間のモーフィングアニメーション作成も可能

load trajectory "Mol1.pdb" "Mol2.pdb"; animation morph 32 32 コマの軌跡を作る。これを動作させて保存する。
 詳細は、<http://chemapps.stolaf.edu/jmol/docs/#capture> や #animation


結合情報の変更

connect delete 既存の結合を消す (左)
 connect デフォルトの結合を生成 (中央)
 connect 1.8 [_C] [_C] 距離 1.8Å 以内の炭素同士のみを結合 (右)
 それぞれ connect 2.2, 2.5, 3.0



C60 を適当に結合してみた例(ファイル c60.fck 等)

connect delete
 connect 3
 color yellow
 color bond red // 左側
 slab on
 slab 60 // 右側 断面表示



結合次数等の強制的な変更 (太すぎる時は wireframe 10 等、表示が変わらない時は set bonds on、CPK 表示で見えないときは cpk off)

connect @1 @2 delete
 connect {atomno=1} {atomno=2} delete 同上
 connect {atomno=1} {atomno=2} single
 connect {atomno=1} {atomno=2} double
 connect {atomno=1} {atomno=2} triple
 その他 上記のオプションの部分、
 partial 1 単結合 実線 single でも同じ
 partial 1.1 単結合 点線
 partial 1.0 五本の点線? (右図下端)
 partial 2 二重結合 実線 double でも同じ
 partial 2.1 同上 実線と点線
 partial 2.2 同上 実線と点線 点線の位置違い
 partial 2.3 同上 点線 2 本
 partial 3 三重結合 実線 triple でも同じ
 partial 3.1 同上 実線と点線
 partial 3.2 同上 実線と点線 点線の位置違い
 partial 3.3 同上 実線と点線 (2 本が点線)
 partial 3.4 同上 実線と点線 点線の位置違い
 partial 3.7 同上 点線 3 本



分子表面に、MMFF94 で与えられる電荷をマッピングする。

水2分子で 2H2O_MMFF94.xyz を使用

`calculate partialCharge; isosurface vdw map mep` // 左側

`calculate partialCharge; isosurface vdw 1.0 map mep` // 右側

後ろに `translucent` をつけると半透明にできる。

さらに 0.5 等で透明度の指定が可能。水1分子のみを選択して実行した場合→

`partialCharge` の代わりに `formalCharge` も可。


色のつけ方は任意に変更可能（色の変化が見えない時は範囲を定める `absolute`）

`calculate partialCharge; isosurface vdw color absolute -0.8 0.3 map mep`

計算して数値として小数点以下2桁で表示する。（; は改行と同じ）


`calculate partialCharge ;wireframe only; label %3.2[partialCharge]`

Gaussian 等での計算結果の電荷を用いることもできる。(cube ファイルを使用)別記



分子表面にある、水が入り込めるポケットを表示する。(1CRN の場合) (load=1crn)
 isosurface Pocket Cavity (左)

分子内部にある、水が入り得るスペースを表示する(右)
 isosurface Interior Cavity




LCAO 軌道を作成して描画 (Linear Combination of Atomic Orbitals)

コンソールから


```
lcaoCartoon create px / py / pz
lcaoCartoon create sp2a/sp2b/sp2c
lcaoCartoon create sp3a/sp3b/sp3c/sp3d
lcaoCartoon delete
lcaoCartoon on/off
```

p 軌道を表示, 右図は pz を描画
 sp2 軌道を表示.
 sp3 軌道を表示.
 軌道を削除する
 軌道表示のオン/オフ



例 ヘモグロビン


```
load==hem set bond off isosurface solvent 1.0 translucent 0.5 color isosurface yellow translucent 0.5 select _Fe ; cpk 80%
```




DNA の残基による色分け例

スクリプトエディタで実行
 すると操作しやすい。
 select all
 cartoon on


```
select G  
color Green  
select C  
color Yellow  
select A  
color Red  
select T  
color Blue
```



```
select all  
cartoon off  
cpk on  
  
select G  
color Green  
select C  
color Yellow  
select A  
color Red  
select T  
color Blue
```




分子中一部の原子
 を選択して、表示モデ
 ルを変更し、色を変更
 することにより、特定
 部位を強調した図な
 どが作成できる



タンパクの表示方法と色のつけ方

```
load=1crn ; cartoon on ; color cartoon structure ; color cartoon shapely ; color cartoon amino
```



下図左から

```
trace on ; color trace structure  
backbone on ; color backbone black  
ribbon on ; color ribbon structure  
meshRibbon on ; color meshRibbon structure  
strand on ; color strand black  
rocket on ; color rocket structure
```



(コイル状部分は α ヘルックス、シート状は β シートを表している。矢印は NHCO の C 末端側を向いている。)

リボン等の詳細な表現方法

```
set hermiteLevel 0 デフォルト値. 10 まで可. リボンを太く、丸くする (リボンのスムージング)  
set ribbonAspectRatio 1 上記を併用するとリボンを丸棒とできる。(リボンの厚さ)
```

複数の表示法を重ねたり、選択部分のみをリボン表示する等も可


```
光の当て方も設定できる(通常の 3D Graphics 技法と同じ)  
set specular 60 光の反射の割合  
set specpower 60 光の反射の強さ  
set ambient 60 間接光の割合  
set diffuse 60 散乱光の割合
```




1CRN の Ramachandran Plot 例 : アミド結合周りの 2つの二面角のプロットによって蛋白のねじれ方を図示する。

メニューの [ファイル] → [PDB を取得] 1CRN

[ファイル] → [コンソール] を開き ramachandran を選択、または [plot ramachandran](#) とキーボードから入力

プロット図を元の表示に戻すには、 ボタン、または右クリック→[モデル]から選択、または [model 1.1](#) 等の元のモデル番号を入力する。



[show RESIDUES](#) 1CRN の場合の残基の表示例 表示された名称で選択できる。(PDB ファイル中で定義されている場合のみ)

[select \[THR\]1:A](#) 、全ての[THR]の選択は、[select \[THR\]](#) 等

[THR]1:A	[SER]6:A	[SER]11:A	[CYS]16:A	[THR]21:A	[CYS]26:A	[GLY]31:A	[PRO]36:A	[PRO]41:A	[ASN]46:A
[THR]2:A	[ILE]7:A	[ASN]12:A	[ARG]17:A	[PRO]22:A	[ALA]27:A	[CYS]32:A	[GLY]37:A	[GLY]42:A	
[CYS]3:A	[VAL]8:A	[PHE]13:A	[LEU]18:A	[GLU]23:A	[THR]28:A	[ILE]33:A	[ALA]38:A	[ASP]43:A	
[CYS]4:A	[ALA]9:A	[ASN]14:A	[PRO]19:A	[ALA]24:A	[TYR]29:A	[ILE]34:A	[THR]39:A	[TYR]44:A	
[PRO]5:A	[ARG]10:A	[VAL]15:A	[GLY]20:A	[ILE]25:A	[THR]30:A	[ILE]35:A	[CYS]40:A	[ALA]45:A	

選択は、右クリック→[選択] → [タンパク質] → [残基の種類で] から可

[show SEQUENCE](#) 配列 下記名称でも選択できる。

```
Model 1
Chain A:
[THR]1 [THR]2 [CYS]3 [CYS]4 [PRO]5 [SER]6 [ILE]7 [VAL]8 [ALA]9 [ARG]10 [SER]11 [ASN]12 [PHE]13 [ASN]14 [VAL]15 [CYS]16
[ARG]17 [LEU]18 [PRO]19 [GLY]20 [THR]21 [PRO]22 [GLU]23 [ALA]24 [ILE]25 [CYS]26 [ALA]27 [THR]28 [TYR]29 [THR]30 [GLY]31 [CYS]32
[ILE]33 [ILE]34 [ILE]35 [PRO]36 [GLY]37 [ALA]38 [THR]39 [CYS]40 [PRO]41 [GLY]42 [ASP]43 [TYR]44 [ALA]45 [ASN]46

Model 1
Chain A:
[THR]2 [CYS]3 [CYS]4 [PRO]5 [SER]6 [ILE]7 [VAL]8 [ALA]9 [ARG]10 [SER]11 [ASN]12 [PHE]13 [ASN]14 [VAL]15 [CYS]16 [ARG]17
[LEU]18 [PRO]19 [GLY]20 [THR]21 [PRO]22 [GLU]23 [ALA]24 [ILE]25 [CYS]26 [ALA]27 [THR]28 [TYR]29 [THR]30 [GLY]31 [CYS]32 [ILE]33
[ILE]34 [ILE]35 [PRO]36 [GLY]37 [ALA]38 [THR]39 [CYS]40 [PRO]41 [GLY]42 [ASP]43 [TYR]44 [ALA]45
```

例


GeTPP の meso 位の炭素原子を強調して表示

GeTPP(O)(OH).xyz を読み込む.(授業用フォルダの molecules から)

meso 位の 4 つの炭素の番号を確かめる。マウスでクリックしてコンソールを見る等. 7,13,19,25 とわかる。

```
cpk // 全体を cpk 表示とする
color "#EEEEEE" // 全体を薄い灰色にする. RGB 各色を順番に 00~FF の 16 進数で入力. #000000 (黒, 光なし) #FFFFFF (白)
// 色指定の詳細は、http://jmol.sourceforge.net/jscolors/
```

```
select atomno=7
select add atomno=13
select add atomno=19
select add atomno=25
または、一度に
select (atomno= 7, atomno=13, atomno=19, atomno=25)
color red // 選択された原子を赤くする
```



結晶構造の作成例

結晶等のデータベースの例

Crystallography Open Database	http://www.crystallography.net/
IMA Database of Mineral Properties	http://rruff.info/ima/
Mineralogy Database	http://www.webmineral.com/

他のソフトウェアの例

Mercury <http://www.ccdc.cam.ac.uk/Solutions/CSDSsystem/Pages/Mercury.aspx>
 英ケンブリッジ大. 高機能でフリー (オープンソースではない) Win, Mac 版あり
 (Jmol で読み込めない CIF ファイルを保存し直す等でも利用)

粘土(モンモリロナイト)の結晶構造の作成の手順

Crystallography Open Database で Montmorillonite を検索して、Crystal Informatin File の 1100106.cif を得る。

授業用フォルダの Montmorillonite_1100106.cif でもよい。

CIF ファイルはテキストエディタで開いて中身を改変することも可能。


```
load "Montmorillonite_1100106.cif" // ファイルの置いてある場所に注意
set defaultDirectory "パス名" // 読み込むディレクトリを指定しておいてもよい
```

3 × 3 × 2 のサイズで読み込む

```
load "Montmorillonite_1100106.cif" {3,3,2}
```

```
zoom 80
rotate y 90
unitcell off
axes off
```


// 表示サイズを 80% に
// Y 軸を 90° 回転させる
// 単位格子, ユニットの表示を消す
// 軸の表示を消す



中央部の 1 層を残し、両側を削除する操作

```
select none
```

// 全原子が選択された状態なので、全て解除する

ウィンドウの上中央から  を選び、シフトキーを押しながら両側の不要な部分を選択する。


```
delete selected
```

// 選択された部分を削除する

選択部分をわかりやすいように黄色く表示するには、


```
selectionHalos on/off // または、右クリック [選択] →  選択対象を暈で囲む
```

必要に応じて、上記の選択、削除を繰り返す。




結合情報が適当でないので、一旦消去する

```
select all
connect delete
```




```
connect 1.0 2.0 (_Si) (_O)
```

// 1.0Å ~ 2.0Å の範囲の Si と O を結合する。




```
connect 1.0 2.5 (_Al) (_O)
connect 1.0 2.5 (_Mg) (_O)
```

// 適宜 1.0Å ~ 2.5Å の範囲の Al や Mg と O を結合する。



```
connect (_Mg) (_O) delete
```

// Mg-O 結合を消したい場合



4 面体を表示する


```
select _Si
polyhedra 4 radius 1.8 translucent 0.2
```

// 中心とする Si 原子を選択する
// 中心から 1.8Å にある原子までで 4 面体を作成する。オプションで透明度を 0.2 とした。

6 面体を表示する。

```
select _Al, _Mg
polyhedra 6 radius 2.4 translucent 0.3
```

// 中心とする Al と Mg 原子を選択する
// 中心から 2.4Å にある原子までで 6 面体を作成する。




不要な原子は、選択して削除する。
面の色は、中心となる原子の色が適用されます。

```
color lightblue
```

// 現在選択されている Al と Mg が薄青になります

Si を黄色に、球をなくし、結合を細く

```
select _Si
color yellow
cpk off
wireframe 10 // 右図
```



まわりの不要部分は適宜削除のこと。

Sumecton SA のように表面の Si の一部を Al にしたい場合には、Jmol のエディタで Si を Al に変更する、


または、該当する Si の番号をクリック等で調べ、その番号の色だけを変える、

または、XYZ 形式で書き出して `write xyz "testJmol.xyz"` この xyz 形式ファイルをテキストエディタで開き、該当する Si を Al に書き換えるなどで対応できます。

ファイル出力は、これでも可 `write properties x y z format "%s %f %f %f" "output"`

Niobate(K4Nb6O17) 結晶構造の作成も同様にできます。


CIF ファイル <http://www.crystallography.net/cod/1001842.html>



```
set axisMode 0/1/2
set axesOffset -0.2
```

0:window, 1:分子, 2:unit cell
軸が unit cell に重なって見づらい場合にずらす

その他、このような表示も可能



```
connect {atomno=1} {atomno=26} delete
contact {atomno=1} {atomno=26} translucent 0.6
contact {atomno=1} {atomno=26} plane translucent 0.6
contact {atomno=1} {atomno=26} connect translucent 0.6
contact {atomno=1} {atomno=26} surface translucent 0.8
contact {atomno=1} {atomno=26} cap translucent 0.2
contact {atomno=1} {atomno=26} frim translucent 0.2
contact {atomno=1} {atomno=26} full translucent 0.6
```

#特定の結合を消す
 #原子間に円盤を描く デフォルト
 #原子間に円盤を描く 同上
 #原子間に筒を描く
 #1番目の原子のvdW球を描く。色は2番目の原子までのグラデーションとする。
 #1番目の原子のvdW球を描く。まわりの原子までのグラデーションとする。
 #surfaceの球の2番目の原子近傍を消した球とする。
 #surfaceの球の2番目の原子近傍のみの皿型とする。
 #結合間に、両者のvdW球の重なった部分のみを描く

```
set radius 2.0
set solvent ON
dots ON
dots 1000%
```

```
select all
dots off
```

操作の概要

メニュー表示は、画面上を右クリック
 平面内移動は、control+右クリック
 拡大/縮小は、Shiftクリックしながら、上下方向にドラッグ またはホイール
 Z軸回りの回転は、Shiftクリックしながら、水平方向にドラッグ

分子軌道の表示をやめる/する

分子軌道のメッシュを細かくする
 分子軌道のメッシュを消す/出す
 分子軌道を曲面で表示する/しない

分子軌道を半透明にする

半透明をやめる

分子軌道のローブのサイズを変える

軌道の色を変える

分子軌道を点で表示する/しない

点が見えるようにメッシュや面の表示を消す必要あり。

軌道の断面表示

断面の位置設定

XY平面上で切った軌道を表示する

同上 指定した面で

同上面を消す

コンソールコマンド

mo off / on

mo resolution 10 等

mo nomesh / mesh

mo fill / nofill

mo translucent 0.6

mo opaque

mo cutoff 0.01 等 0.05 がデフォルト

mo color yellow purple

mo dots / nodots

slab on / off

slab 60 等


mo plane "xy"

mo plane "xz"

mo plane "yz"

mo plane {0 0 1}

mo noplane



画像ファイルに書き出す

write image "ファイル名.jpg"

右クリック→[ファイル]→[エクスポート]→[JPG イメージを書き出す] 等


構造最適化の途中の座標をファイルに書き出すには、model n で選択し、画面表示した後、 ([Refresh] ボタンが必要な場合もあります) メニューから [ツール] → [Gaussian]

その他の例 set リターン で現在の(変更可能な)設定が全て表示されます。

現在の表示状態を再現するスクリプトを得るには、

左から5番目のボタン [現在の状況を書き出し]

または、メニューの[ファイル]→[エクスポート]→[状態を書き出し]. からスクリプトとして保存することができます。(かなりの行数) スクリプトエディタ等で実行すると再現できます。



Avogadro では、収束構造が表示されます。

Version 120 では軌道を表示できない場合があります。110 は可 <https://sourceforge.net/projects/avogadro/files/avogadro/>

Jmol の場合

ログファイルを Jmol のウィンドウにドロップして読み込むと Jmol のコンソールに 10 models

4 molecular orbitals in model 1.2 初期構造の分子軌道が4つ

4 molecular orbitals in model 1.10 収束構造の分子軌道が4つ

などと表示されます。

10 models もの構造情報を持っている理由は、log ファイル中の座標系の異なる 2 種の Input orientation:(入力された座標系)と Standard orientation:(分子の重心を中心とした座標系)の両方を読み込み、さらに収束後の構造の出力も読み込まれているからです。これは、右クリック→モデルからも見ることができます。

(初期値と収束構造を含むループ 4 回+収束構造 1) × 2 = 10

コンソールから

ボタンによる操作も可




model 1

model 10

等で表示でき、収束途中の構造を観察できます。

model 0 では、全てが重ねて表示されます。



コンソールの下[エディタ]または、メニューの[ファイル]から[スクリプトエディタ]を開き、

background white

cpk 15%

wireframe 15

#set antiAliasDisplay true

set antiAliasDisplay false

をコピー&ペーストして[実行]してください。

収束構造の分子軌道が必要なので

model 1.10

と入力して下さい。

右クリック→[面]→[分子軌道]→ でも可

H2O.log の場合は、

model 1.10 と入力してリターン

以下、コンソールに入力して軌道を観察してください。

mo 1

mo 2

mo 3

mo 4

空軌道が 3 つ(2,3,4)なる理由は、与えた基底関数の 6-31G(d)では、1 つの 1s 軌道に 2 つの波動関数が割り当てられるため水素分子では計 4 つの波動関数となり、内 1 つが被占軌道となって残り 3 つが空軌道となります。一番小さい基底関数の STO-3G では HOMO と LUMO だけが計算されます。

HOMO,LUMO 周辺の軌道は、

mo homo

mo homo-1 #H2 の場合は存在しないのでエラー


mo lumo

mo lumo+1

軌道を前後するには、

mo next

mo previous



ラジカル種の計算とスピン分布図の描画

Gaussian のインストールされているマシンには、**formchk コマンド**が入っています。これを使ってチェックポイントファイルから拡張子が .fchk の **フォーマットド・チェックポイントファイル**を生成します。これはテキストファイルです。

```
formchk CH3r_B6.chk
```

これを実行すると、**CH3r_B6.fchk** が生成します。

さらにこの fchk ファイルから **cubegen コマンド**を使ってグラフィックス表示のための cube ファイルを生成します。

```
cubegen 0 spin CH3r_B6.fchk OHR_B6.cube 0
```

Jmol によってこの cube ファイルからスピン分布図を作ります。

Jmol はファイル名だけを与えた場合、cube ファイルの置いてあるディレクトリ(場所)を認識できません。そのため、ファイルの置いてある場所を下記のコマンドで指定します。

ファイルのフルパスの取得法

Windows では、cube ファイルのアイコンを「Shift を押しながら右クリック」します。

macOS では、cube ファイルのアイコンを option+command+C または、ターミナル画面にドロップ。

上記の cube ファイルへのフルパスからファイル名を除いて、ディレクトリ部分だけにして下記のコマンドを作成してください。

スクリプトエディタに、下記のコマンドを置いて実行して下さい。

```
set defaultDirectory "cube ファイルの置いてあるディレクトリのフルパス"
```

ファイル名まで入れないこと

下記を Jmol のスクリプトエディタに入力して実行してください。

```
set DefaultDirectory "???" ログファイルのあるディレクトリを指定
```

Windows では、最後に*または\を入れないでください。


```
load "CH3r_B6.log" ログファイルを読み込む
```

```
wireframe 10
```

```
cpk 10%
```


```
background white
```

```
frame LAST
```

 の機能

```
isosurface s1 cutoff 0.02 color red CH3r_B6.cube
```

```
isosurface s2 cutoff -0.02 color blue CH3r_B6.cube
```




ローブを半透明にするには、最後の2行を変更してください。

半透明にすると影がなくなるため立体感がなくなります。

```
isosurface s1 cutoff 0.02 color red CH3r_B6.cube translucent 0.8
```

```
isosurface s2 cutoff -0.02 color blue CH3r_B6.cube translucent 0.8
```




ローブをメッシュにするには、

```
set antialiasDisplay FALSE メッシュの場合は平滑化を off にした方がよい
```

```
isosurface s1 cutoff 0.02 color red CH3r_B6.cube mesh nofill
```

```
isosurface s2 cutoff -0.02 color blue CH3r_B6.cube mesh nofill
```




cutoff 値を変更してローブの大きさを変えることもできます。

大きくすると水素にもスピンの少しあるのが見えてきます。

```
isosurface s1 cutoff 0.001 color red CH3r_B6.cube
```

```
isosurface s2 cutoff -0.001 color blue CH3r_B6.cube
```




色は、自由に設定できます。

このページの末尾参照 <https://jmol.sourceforge.net/jscolors/>

色は通常、黄～緑が使われているようです。

```
isosurface s1 cutoff 0.02 color aquamarine CH3r_B6.cube
```

```
isosurface s2 cutoff -0.02 color aquamarine CH3r_B6.cube
```



ここで、s1, s2 は表面に仮につけた名前で、任意の名称が可能です。

多くの場合 α 電子が対電子となるので、 β 電子(青色)が対電子として表示されることはありません。

Jmol のグラフィックス表示の詳細設定について

コンソールからコマンド

```
$ set LIGHTING
```

と入力すると、分子模型への光の当て方などの詳細設定が下記のように表示されます。

```
set ambientPercent 45;
set diffusePercent 84;
set specular true;
set specularPercent 22;
set specularPower 40;
set specularExponent 6;
set celShading false;
set celShadingPower 10;
set zShadePower 3;
set zDepth 0;
set zSlab 50;
set zShade false;
```

それぞれの値について

```
set ambientPercent 60
```

などと値を変更することによって、グラフィックス表示を詳細に変更することができます。

マニュアルは、下記の set (lighting) を参照

<https://chemapps.stolaf.edu/jmol/docs/#setlighting>

IR


HULINKS 社の「Gaussian 可視化のためのチュートリアル:GaussView Version: 3.09」

「GaussView を利用した振動数(Vibrational Frequencies)の可視化」

https://www.hulinks.co.jp/support/gaussian/tutorial_gv.html#g_v_05

のページにあったログファイル **DMACET.log** を用います。計算は B3LYP/6-31G(d) 水溶媒中で行われています。


なお、この計算結果は構造最適化が完全に収束しておらず、虚の振動数が表れているため、振動構造解析時結果として用いるには不適當です。



Avogadro で

ログファイルを Avogadro で開くと右側に振動数が表示され、右下でベクトル表示やアニメーション表示ができます。

[スペクトルを表示]は macOS 版では機能しません。Windows 版は？



Jmol では、グラフィックスやアニメーションの詳細な設定ができます。

Jmol では、Gaussian03 で計算されたの **DMACET.log** が読み込めなかったので、別途計算した **DMACET 1 .log** を読み込んでください。

振動開始/停止 [vibration on/off](#)

振動モードは、ウィンドウの左下に表示されます。

振動モードの選択は、画面を右クリック→モデル選択 または、 または、**model** 数字。 [model next](#)、[model previous](#) も可

1秒間の振動数を与える [vibration](#) 数字 0.5 は1秒に2回振動

振動の動きの大きさをスケールで与える [vibration SCALE](#) 数字 2 で2倍


ベクトル矢印の表示 [vector on/off](#)

矢印の太さを変える [vector](#) 数字

矢印の長さを変える [vibration MAX](#) 5

矢印の色を変える [color vector blue](#)

ベクトルをスケールで与える [vector SCALE](#) 数字



アニメーション中

回転させながら眺めるには、

Z 軸で回転は、 [spin on/off](#)

軸を表示する [spin z 100](#)

軸の太さを変更 [axes on/off](#)

軸の長さを変更 [axes 10](#)

アニメーションを GIF アニメーションファイルにすることもできます。 [axes scale 10](#)




```
wireframe 100
color white

select [PL9]
wireframe 100
color lightblue

select [TYR]161
wireframe 100
color yellow
```

一度に複数回 select してから表示を変更する場合には、追加選択の select add を使う。

参考：「人工光合成とは何か」井上晴夫監修 ブルーボックス
PDBj 今月の分子 59 光化学系 II (Photosystem II) <https://pdbj.org/mom/59>



歪んだ椅子と水 4 分子

印刷用には、背景色を白(background white)として、分子の色を変更すると良い。

Trace 表示(左)と Cartoon 表示(右)




左
select all
trace on
color trace structure
wireframe off
cpk off

structure の代わりに shapely や amino も可
結合表示を消す
原子の球を消す

右
trace off
cartoon on
color cartoon structure

Trace 表示を消す




等


Photosystem I の 1JB0.pdb の場合 0 はゼロ、o オーとすると別の分子。

参考
PDBj 今月の分子 22 光化学系 I (Photosystem I) <https://pdbj.org/mom/22>
「人工光合成とは何か」の 87 頁 図 2-18 と同様

[CLA]と[SF4]を表示



[CLA]を表示



クロロフィル a と Cartoon 表示



Gaussian の構造最適化の初期構造から収束構造までをアニメーションで見ると、

log ファイルを読み込む。 画面にドロップ or load
モデル数を確認する 20 models の場合

```
animation frames [1-20]      #アニメーションにするフレームの指定
animation fps 4              # 1 秒のコマ数
animation on                  #アニメーション開始
さらに
animation mode once          #ループしない デフォルト
animation mode loop          #アニメーションをループする
animation MODE PALINDROME    #行って戻ってのループ
```

capture して GIF ファイルにして、その後、

GIF アニメを MP4 に変換する. (intel mac で動作)

```
ffmpeg -i input.gif -movflags faststart -pix_fmt yuv420p -vf "scale=trunc(iw/2)*2:trunc(ih/2)*2" output.mp4
https://qiita.com/razokulover/items/fe45833c997b338c638a
```

getProperty modelInfo

の出力の

```
modelInfo.models[?].num
```

が、モデル数を返す

```
getProperty modelInfo.models[13].num
```

が、

```
modelInfo.models[13].num          13
```

とモデル数を返す。が、13 を取り出す方法は？

```
getProperty modelInfo.modelsSelected          [0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12]
```

#SSA3 1MV 用

```
background white
set perspectiveDepth false
select all
connect delete
connect 1.0 2.0 (_Si) (_O)
connect 1.0 2.5 (_Al) (_O)
connect 1.0 2.5 (_Mg) (_O)
connect 0.5 1.0 (_H) (_O)
connect 1.0 2.0 (_C) (_C)
connect 1.0 2.0 (_C) (_N)
connect 0.5 1.0 (_H) (_C)
```

```
cpk 10%
wireframe 10
```

```
select atomno>=626
cpk 100%
```

```
select _Al and atomno<=473
cpk 40%
color red
select _Al and atomno>473
cpk 25%
color blue
```

スクリプト

変数値をコマンドで使うには、@変数 とする

for 文と if 文の例は、

Gaussian のログファイルを読み込み、model を選択して収束構造を表示してから HOMO,LUMO まわりの 15 の MO の画像ファイルを自動出力する。

ho に HOMO 起動の番号を入力しておく。

```
for (i= ho-10 ; i<= ho+5 ; i++)
  f="画像ファイル名"
  if (i<10) {
    f=f+"000"+i+".jpg"
  } else if (i<100) {
    f=f+"00"+i+".jpg"
```

```
} else if (i<1000) {  
  f=f+"0"+i+".jpg"  
} else {  
  f=f+i+".jpg"  
}  
ff=i  
mo @i  
write image @f  
end for
```

IR, NMR, Mass の計算結果ログファイルを Jmol に読み込んで show NMR 等のコマンドで JspecView でグラフ化できる。

岡山大学 JspecView で IR スペクトル表示

<https://www.chem.ous.ac.jp/~waka/spectra/ir-spectra/index2.php?filename=dibenzylideneacetone.jdx&startindex=209&endindex=3735>